Calculus Single Variable
Brian E. Blank and Steven G. Krantz

Section 6.5
Applications of the Exponential Function
Part III

Load

1. The Basic Differential Equation

If A is any constant (usually positive in applications) and if k is any nonzero constant, then the unique solution of the initial value problem

$$\frac{d}{dt}\,y(t) = k\,y(t) \quad , \quad y(0) = A$$

is

$$y(t) = A\,e^{(k\,t)}.$$

2. A Generalization of the Basic Differential Equation

Many differential equations that arise in practice are related to the differential equation for exponential growth/decay.

<u>Theorem</u>: The unique solution to the initial value problem

$$\frac{d}{dt}\,u(t) = k\,u(t) + b, \quad u(0) = A$$

is

$$u(t) = \left(A + \frac{b}{k}\right) e^{(k\,t)} - \frac{b}{k} \quad .$$

Proof:

```
> ode := diff(u(t), t) = k*u(t) + b;
```

$$ode := \frac{d}{dt}\,u(t) = k\,u(t) + b$$

```
> eqn1 := u(t) = w(t) - b/k;
  eqn2 := w(t) = solve(eqn1, w(t));
```

$$eqn1 := u(t) = w(t) - \frac{b}{k}$$

$$eqn2 := w(t) = \frac{k\,u(t) + b}{k}$$

```
> ic1 := u(0) = A;
  eqn3 := subs(t = 0, eqn2);
  ic2 := subs(ic1, eqn3);
```

$$ic1 := u(0) = A$$

$$eqn3 := w(0) = \frac{k\,u(0) + b}{k}$$

$$ic2 := w(0) = \frac{b + A\,k}{k}$$

```
> ode2 := subs(eqn1, ode);
```

$$ode2 := \frac{\partial}{\partial t}\left(w(t) - \frac{b}{k}\right) = k\left(w(t) - \frac{b}{k}\right) + b$$

```
> ode3 := simplify(ode2);
```

$$ode3 := \frac{d}{dt}\,w(t) = w(t)\,k$$

```
> soln1 := w(t) = w(0)*exp(k*t);
```

$$soln1 := w(t) = w(0)\,\mathbf{e}^{(k\,t)}$$

```
> soln2 := subs({eqn2, ic2}, soln1);
```

$$soln2 := \frac{k\,u(t) + b}{k} = \frac{\mathbf{e}^{(k\,t)}\,(b + A\,k)}{k}$$

```
> soln := u(t) = expand(solve(soln2, u(t)));
```

$$soln := u(t) = -\frac{b}{k} + \frac{\mathbf{e}^{(k\,t)}\,b}{k} + \mathbf{e}^{(k\,t)}\,A$$

We can also allow Maple to obtain this result unaided:

```
> IVP := {diff(u(t),t) = k*u(t)+b, u(0)=A};
```

$$IVP := \{ \frac{d}{dt} u(t) = k\, u(t) + b, u(0) = A \}$$

```
> dsolve(IVP,u(t));
```

$$u(t) = -\frac{b}{k} + \frac{e^{(k\,t)} (b + A\,k)}{k}$$

3. Limiting Behaviour of Solution

If $k < 0$ then the solution to the initial value problem $\frac{d}{dt} u(t) = k\, u(t) + b, \quad u(0) = A$ has a finite limit as $t \to \infty$.

```
> limit(exp(k*t)*(b+A*k)/k - b/k, t=infinity)  assuming k < 0;
```

$$-\frac{b}{k}$$

In other words, $y = -\frac{b}{k}$ is a horizontal asymptote of the graph of $y = u(t)$.

4. Newton's Law of Heating/Cooling

At time $t = 0$ an object at temperature T_0 is placed in an environment that has constant temperature T_∞. Let $T(t)$ denote the temperature of the object at time t. Then $T(t)$ satisfies the initial value problem

$$\frac{d}{dt} T(t) = K\,(T_\infty - T(t)) \quad , \quad T(0) = T_0 \ .$$

This is the equation of Section 2 above with $u(t) = T(t)$, $k = -K$, $b = K\,T_\infty$., and $A = T_0$.

```
> subs( {u(t) = T(t), k = -K, b = K*T[infinity], A = T[0]},
  soln);
```

$$T(t) = T_\infty - e^{(-K\,t)}\, T_\infty + e^{(-K\,t)}\, T_0$$

Exercise

(Exercise 14 Page 472)

A very hot pan is placed in hot dish water. The temperature difference is 300 degrees at 1:00 PM and 150 degrees at 1:06 PM. When is the difference in temperature equal to 100 degrees?

Solution

Since the data for this problem is given in terms of the differences $T(t) - T_\infty$, it is covenient to write the solution as

$$\mathrm{T}(t) - T_\infty = \mathbf{e}^{(-K\,t)}\,(T_0 - T_\infty)$$

Let us set the origin of the time axis at 1:00 PM. Then the information given for that time tells us that $T_0 - T_\infty = 300$. Thus

$$\mathrm{T}(t) - T_\infty = 300\,\mathbf{e}^{(-K\,t)}$$

The information at 1:06 PM tells us that $\mathrm{T}(6) - T_\infty = 150$. But $\mathrm{T}(6) - T_\infty = 300\,\mathbf{e}^{(-6\,K)}$. Thus

$$300\,\mathbf{e}^{(-6\,K)} = 150 .$$

We solve

```
> eqn1 := K = solve( 300*exp(-6*K) = 150 , K );
```

$$eqn1 := K = \frac{1}{6}\ln(2)$$

```
> eqn2 := T(t)-T[infinity] = 300*exp(-K*t);
```

$$eqn2 := \mathrm{T}(t) - T_\infty = 300\,\mathbf{e}^{(-K\,t)}$$

```
> eqn3 := subs(eqn1, eqn2);
```

$$eqn3 := \mathrm{T}(t) - T_\infty = 300\,\mathbf{e}^{(-1/6\,\ln(2)\,t)}$$

```
> answer := solve(rhs(eqn3) = 100, t);
```

$$answer := \frac{6\ln(3)}{\ln(2)}$$

```
> evalf(answer);
```

$$9.509775006$$

```
> answer := 9*minutes + 0.509775006*60*seconds*`  after 1 PM`;
```

$$answer := 9\ minutes + 30.58650036\ seconds\ \ after\ 1\ PM$$

The temperature of the pot will continue to approach the temperature of the surrounding dish water. In other words, the difference tends to 0. Here is a plot

```
> plot(300*exp(-1/6*ln(2)*t), t = 0 ..60);
```



5. Linear Drag Law

The acceleration of an object of mass *m* due to gravity and linear drag is given by

$$\frac{d}{dt}\mathrm{v}(t) = -\left(\frac{K\,\mathrm{v}(t)}{m} + g\right)$$

for some positive constant K. (In this equation, which has a minus sign applied to each summand on the right side, we take the gravitational acceleration g to be a *positive* constant.)

This is the equation of Section 2 above with $\mathrm{u}(t) = \mathrm{v}(t)$, $k = -\frac{K}{m}$, $b = -g$., and $A = v_0$.

The solution is:

```
> subs( {u(t) = v(t), k = -K/m, b = -g, A = v[0]}, soln);
```

$$v(t) = -\frac{g\,m}{K} + \frac{m\,\mathbf{e}^{\left(-\frac{K\,t}{m}\right)}\,g}{K} + \mathbf{e}^{\left(-\frac{K\,t}{m}\right)}\,v_0$$

Notice that

$$\lim_{t \to \infty} v(t) = -\frac{mg}{K}.$$

This quantity is called the ***terminal velocity***. The terminology is a bit misleading: this speed is approached but not attained.

Exercise (Exercise 61 Page 476)

Photo # NH 94785 USS Arizona in the East River, New York City, circa mid-1916

On December 7 1941, during the attack on Pearl Harbor, an 800 kg bomb was dropped from a Nakajima BN52 Kate bomber flying at an altitude of 3170 meters. The bomb struck the battleship ***USS Arizona***, igniting its black powder magazine which in turn set off a series of catastrophic explosions. The ship sank in nine minutes with a death toll of 1177.

Photo # USN 1021538 Burned-out wreck of USS Arizona, at Pearl Harbor, December 1941



An analysis of the attack that was published in 1997 asserted
that the flight of the bomb lasted 26 seconds. Accept that figure.
Assume the Linear Drag Law $\mathrm{R}(v) = -K\,v$..

a) What was the value of K ?

Solution

Let $\mathrm{y}(t)$ and $\mathrm{v}(t)$ be the height in meters and velocity in meters/second, of the bomb t seconds after it was dropped.

```
> v := t -> subs({m = 800, g = 9.81, v[0]=0},
  -g/K*m+1/K*m*exp(-K/m*t)*g+exp(-K/m*t)*v[0]);
```

$$v := t \rightarrow \mathrm{subs}\left(\{m = 800, g = 9.81, v_0 = 0\}, -\frac{g\,m}{K} + \frac{m\,\mathbf{e}^{\left(-\frac{K\,t}{m}\right)}\,g}{K} + \mathbf{e}^{\left(-\frac{K\,t}{m}\right)}\,v_0\right)$$

```
> 'v(t)' = v(t);
```

$$\mathrm{v}(t) = -\frac{7848.00}{K} + \frac{7848.00\,\mathbf{e}^{\left(-\frac{K\,t}{800}\right)}}{K}$$

```
> y := t -> 3170 + int(v(s), s = 0 .. t);
```

$$y := t \to 3170 + \int_0^t \mathrm{v}(s)\, ds$$

Since the bomb hit the USS Arizona 26 seconds after it was dropped, we have

```
> eqn := y(26) = 0;
```

$$eqn := 3170 - \frac{15696.\,(13.\,K + 400.\,\mathbf{e}^{(-0.03250000000\,K)} - 400.)}{K^2} = 0$$

```
> eqn_for_K := K = fsolve(eqn, K, 0 .. infinity);
```

$$eqn\_for\_K := K = 4.197705090$$

b) What was the theoretical terminal velocity of the bomb?

```
> limit(subs(eqn_for_K, v(t)), t = infinity);
```

$$-1869.592988$$

c) What was the actual velocity of the bomb when it struck the USS Arizona. (State your answer in meters per second but also convert to miles per hour.)

```
> evalf(subs(eqn_for_K, v(26)))*meters_per_second;
```

$$-238.426594\ meters\_per\_second$$

```
> 3600*convert(-238.426594, units, m/s, mi/s)*miles_per_hour;
```

$$-533.3451012\ miles\_per\_hour$$

A plot of shows that the magnitude of the velocity is growing close to linearly. Air drag had little effect.

```
> plot(subs(eqn_for_K, v(t)), t = 0 .. 26);
```



d) Plot the height of the bomb for $0 \le t \le 26$.

Solution:

```
> plot( subs(eqn_for_K, y(t)), t = 0 .. 26);
```



e) What would the time of fall have been had there been no air drag?
The velocity on impact?

Solution:

With no drag the distance the bomb had dropped after t seconds would have been $\frac{g\,t^2}{2}$.
In this case, time to impact would have been

```
> T := fsolve(subs(g = 9.81, g*t^2/2 = 3170), t, 0 .. infinity);
```

$$T := 25.42202405$$

(About half a second less than the actual fall.)

The velocity on impact would have been

```
> subs(g = 9.81, -g*T);
```

$$-249.3900559$$

(About 10 m/s faster than the actual final velocity.)

Exercise
(Exercise 66 Page 477)

As of this writing, former president George Herbert Bush has been the only American president to have jumped from an airplane. During World War II, the future president parachuted from his airplane when it was shot down.

On March 25 1997, at the age of 72, the former president made a recreational skydive

and parachute jump.

Assume that the mass of the president and his jumping gear was 7.03 lb s^2/ft.

Assume the Linear Resistance Law:

$$R(v) = -1.4\,v$$

during the skydive phase of the jump and

$$R(v) = -\kappa\,v$$

after deployment of the parachute. In this formula κ is an unknown constant that is to be determined.

a) The president jumped from a height of 12500 feet and did not open his parachute until he reached 4500 feet.

- Compute the duration τ of this skydive.
- What was the president's velocity $v(\tau)$ when he opened his parachute?
- Plot $v(t)$ for for $0 < t < \tau$.

Solution

```
> v1 := t -> subs({m = 7.03, K = 1.4, g = 32, v[0]=0},
-g/K*m+1/K*m*exp(-K/m*t)*g+exp(-K/m*t)*v[0]);
```

$$v1 := t \rightarrow \text{subs}\left(\{ v_0 = 0, m = 7.03, K = 1.4, g = 32 \}, -\frac{g\,m}{K} + \frac{m\,\mathbf{e}^{\left(-\frac{K\,t}{m}\right)}\,g}{K} + \mathbf{e}^{\left(-\frac{K\,t}{m}\right)}\,v_0 \right)$$

```
> 'v1(t)' = v1(t);
```

$$\text{v1}(t) = -160.6857143 + 160.6857143\,\mathbf{e}^{(-0.1991465149\,t)}$$

Let $\text{y1}(t)$ denote President Bush's height above the ground during the skydive portion of his jump:

```
> y1 := t -> 14500 + int(v1(s), s = 0 .. t);
```

$$y1 := t \rightarrow 14500 + \int_0^t \text{v1}(s)\,ds$$

```
> tau := fsolve(y1(t) = 4500, t, 0 .. infinity);
```

$$\tau := 67.25470685$$

```
> v1(tau);
```

$$-160.6857143 + 160.6857143\,\mathbf{e}^{(-13.39354048\ seconds)}$$

```
> plot(v1(t), t = 0 .. tau, tickmarks=[6,4]);
```



b) The total time of the president's dive and jump was about 540 seconds.
Determine κ to three significant digits.

Solution

Let $y2(t)$ denote President Bush's height above the ground t seconds after he deployed his parachute.
Let be $v2(t)$ denote his speed t seconds after he deployed his parachute.

```
> v2 := t -> subs({m = 7.03, K = kappa, g = 32, v[0]=v1(tau)},
  -g/K*m+1/K*m*exp(-K/m*t)*g+exp(-K/m*t)*v[0]);
```

$$v2 := t \to \mathrm{subs}\left(\{v_0 = \mathrm{v1}(\tau), K = \kappa, m = 7.03, g = 32\}, -\frac{g\,m}{K} + \frac{m\,\mathbf{e}^{\left(-\frac{K\,t}{m}\right)}\,g}{K} + \mathbf{e}^{\left(-\frac{K\,t}{m}\right)}\,v_0\right)$$

```
> y2 := t -> 4500 + int(v2(s), s= 0 .. t);
```

$$y2 := t \to 4500 + \int_0^t \mathrm{v2}(s)\,ds$$

We are given that $y2(540 - \tau) = 0$. Thus,

```
> eqn_for_kappa := kappa = fsolve(y2(540-tau) = 0, kappa);
```

$$eqn\_for\_kappa := \kappa = 23.86936554$$

c) According to this model, what was the terminal velocity of the president's parachute jump? With what velocity did he touch down?

Solution:

```
> limit(subs(eqn_for_kappa, v2(t)), t = infinity); #terminal velocity
```

$$-9.424632574$$

```
> simplify(subs(eqn_for_kappa, v2(540-tau)));
```

$$-9.424632574$$

d) Plot President Bush's velocity during his jump.

Solution:

```
> v := t -> if t < tau then  v1(t) else subs(eqn_for_kappa, v2(t-tau)) end if;
```

v := **proc**(t)

option *operator*, *arrow*;

if $t < \tau$ **then** v1(t) **else** subs(*eqn_for_kappa*, v2($t - \tau$)) **end if**

end proc

```
> plot(v, 0..540);
```



5. The Hill-Keller Theory of Sprinting

Since the 1920's, mathematical models have been developed to study running performance. These models rely on physical theory, empirical evidence, and statistical analysis. In 1973, J.B. Keller modified an existing mathematical theory of running to make it more appropriate for sprints. Let $v(t)$ denote the speed of a runner at time t. Let $p(t)$ denote the horizontal component of the propulsive force per unit mass that is exerted by the runner at time t. Let $R(v)$ denote the total resistive force per unit mass when the runner's speed is v. We will assume that $R(v) = \frac{v}{\tau}$ for some positive constant τ (the dimension of which is time). We also suppose that $p(t) = P$ for a positive constant P. Obviously this assumption on $p(t)$ is not tenable for races longer than a sprint. With these assumptions, the initial value problem for the runner's speed is:

$$\frac{d}{dt} v(t) = P - \frac{v(t)}{\tau}, \qquad v(0) = 0.$$

From Section 7 above, we have

```
> eqn := factor(subs({u(t)=v(t), A=0, b=P, k=-1/tau},
                     u(t) = (A+b/k)*exp(k*t)-b/k));
```

$$eqn := v(t) = -P\,\tau\left(e^{\left(-\frac{t}{\tau}\right)} - 1\right)$$

The maximum velocity is defined to be:

```
> V[M] = limit(rhs(eqn), t=infinity)  assuming tau>0;
```

$$V_M = P\,\tau$$

We can rewrite the formula for $v(t)$ in terms of V_M:

```
> eqn2 := v(t) = V[M]*(1-exp(-t/tau));
```

$$eqn2 := v(t) = V_M\left(1 - e^{\left(-\frac{t}{\tau}\right)}\right)$$

We can now solve for the distance $x(t)$ the sprinter has run in t seconds:

```
> eqn3 := dsolve({diff(x(t),t) = rhs(eqn2), x(0)=0}, x(t));
```

$$eqn3 := x(t) = -V_M\left(-\tau\, e^{\left(-\frac{t}{\tau}\right)} - t\right) - V_M\,\tau$$

Let us gather terms as follows:

```
> eqn4 :=x(t) = V[M]*(t-tau(1-exp(-1/tau*t)));
```

$$eqn4 := x(t) = V_M\left(t - \tau\left(1 - e^{\left(-\frac{t}{\tau}\right)}\right)\right)$$

One of the great rivalries in track history took place in the 1980s between Carl Lewis (U.S.) and Ben Johnson (Canada).
At the 1987 World Championships in Rome, split times for the Men's 100 Meters competition show that Ben Johnson (Canada) and Carl Lewis (United States) each had maximum velocity V_M equal to 11.8 m/s. Johnson's winning time was 9.83 s and Lewis's second place time was 9.93 s.

Carl Lewis (2'nd) Ben Johnson (1st)

Here are values of P and τ for Ben Johnson and Carl Lewis:

```
> JohnsonParameters := fsolve( {P*tau = 11.8, 100 = subs({t=9.83,
  V[M]=11.8}, rhs(eqn4))}, {P, tau});
```

$$JohnsonParameters := \{ P = 8.705764662, \tau = 1.355423729 \}$$

```
> LewisParameters := fsolve( {P*tau = 11.8, 100 = subs({t=9.93,
  V[M]=11.8}, rhs(eqn4))}, {P, tau});
```

$$LewisParameters := \{ P = 8.107604518, \tau = 1.455423729 \}$$

Here is a plot of their speeds. (Johnson in navy, Lewis in gold)

```
> plot([11.8*(1-exp(-t/1.355423729)),11.8*(1-exp(-t/1.455423729))
  ], t = 0 .. 9.83, color = [NAVY, GOLD]);
```



```
> plot([subs(tau = 1.355423729,
  11.8*(t-tau*(1-exp(-t/tau)))),subs(tau
  =1.455423729,11.8*(t-tau*(1-exp(-t/tau))))], t = 0 .. 9.83,
  color = [NAVY, GOLD]);
```



The code below generates an animation of the race. It can be executed in the Maple worksheet but is of no use in the pdf format.

```
> for j from 1 to 100 do
  comp[j] := plot([[x,2,x=0..subs({t=j*9.83/100, tau
  =1.355423729},11.8*(t-tau*(1-exp(-t/tau))))],[x,1,x=0..subs({t=
  j*9.83/100, tau
  =1.455423729},11.8*(t-tau*(1-exp(-t/tau))))]],color=[NAVY,GOLD]
  ):
  od:

> plots[display](seq(comp[j],j=1..100), insequence = true,
  view=[0..100,0..2]):
> winningMargin := evalf(100 - subs({t=9.83, tau
  =1.455423729},11.8*(t-tau*(1-exp(-t/tau)))));
```

$$winningMargin := 1.159972525$$

Copyright and Author Information

Worksheet Title: BlankKrantz-06_5c-R8.mws   A Maple Release 8 worksheet.

Author: Brian E. Blank
Date Created: 30 September 2006
Date Last Revised: 1 October 2007

For more information, please contact the author:

Department of Mathematics,
Washington University in St. Louis
St. Louis, MO   63130

Telephone: (314) 935-6763
e-mail: brian@math.wustl.edu